
Лабораторная работа №2
Типизированный файл, записи, перечисляемый тип
(2 ак.часа)
Задание. В типизированном файле, хранящем записи вида С1, найти (и вывести) С2, и переставить её по правилу С3 (условия С1, С2 и С3 – см. в таблице ниже).
Для создания типизированного файла предусмотреть две возможности: 
а) считывание данных из заранее подготовленного нестандартного текстового файла, б) добавление данных в типизированный файл в режиме диалога.

Предусмотреть также возможность просмотра данных из типизированного файла.

Для написания своей программы воспользуйтесь примером программы, приведенным ниже. Для выбора нужного действия в программе использовано меню.

Варианты задания

В столбце Тип: п – перечисляемый, с – строковый, ч – числовой, * – по вашему выбору.
*Дату можно представить разными способами: одной строкой ('31.12.2015' или '2015/12/31' и т.п.), тремя целыми числами, двумя числами и перечисляемым типом месяца, типом TDataTime.

	№
	Тип
	С1
	С2
	С3

	1
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Первую запись о выбранном виде транспорта с наименьшим количеством остановок до метро
	С k-ой записью файла

	2
	с
п
п
ч
	Группа (например, А-07-14, ЭР-1-14 и т.д.)

Предмет (физика, матан и т.д.)

Оценка (отл, хор, удовл, неуд)

Количество студентов (из указанной Группы с указанной Оценкой по Предмету)
	Первую запись о группе с наименьшим количеством неудовлетворительным количеством оценок по выбранному предмету
	С последней записью файла

	3
	п
с
*
	Существо (вампир, оборотень и т.д.)

Имя

Дата рождения
	Последнюю запись с самым старым вампиром
	Перед первой записью файла (со сдвигом)

	4
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Последнюю запись о выбранном виде транспорта с наибольшим количеством остановок до метро
	С первой записью файла

	5
	с
ч
п
ч
	Событие

Число

Месяц (январь, февраль и т.д.)

Год
	Первую запись выбранного месяца с событием, содержащим слово «экзамен» в любом регистре


	С k-ой записью файла

	6
	п
с
ч
*
	Тип (мобильный, городской, факс и т.д)

Телефонный номер

Сумма платежа

Дата платежа
	Последнюю запись с выбранным типом связи и суммой платежа не менее 500
	После k-ой записи файла (со сдвигом)



	№
	Тип
	С1
	С2
	С3

	7
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Первую запись о выбранном виде транспорта с буквой «Б» в номере маршрута
	С последней записью файла

	8
	с
п
п
ч
	Группа (например, А-07-14, ЭР-1-14 и т.д.)

Предмет (физика, матан и т.д.)

Оценка (отл, хор, удовл, неуд)

Количество студентов (из указанной Группы с указанной Оценкой по Предмету)
	Последнюю запись о группе без цифры 4  в названии группы 
	После последней записи файла (со сдвигом)

	9
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Последнюю запись о выбранном виде транспорта с двумя единицами в номере маршрута
	Со следующей записью файла

	10
	п
с
*
	Существо (вампир, оборотень и т.д.)

Имя

Дата рождения
	Последнюю запись с самым молодым оборотнем
	Перед k-ой записью файла (со сдвигом)

	11
	с
п
п
ч
	Группа (например, А-07-14, ЭР-1-14 и т.д.)

Предмет (физика, матан и т.д.)

Оценка (отл, хор, удовл, неуд)

Количество студентов (из указанной Группы с указанной Оценкой по Предмету)
	Последнюю запись о группе с двумя единицами в названии группы с не менее чем двумя отличными оценками по любому предмету
	С первой записью файла

	12
	с
п
с
	Преподаватель

Занятие (лекция, лаб, сем, экзамен и т.д.)

Аудитория (например, Ж-113, А-312 и т.д)
	Первую запись с введенной аудиторией и фамилией препода-вателя на букву «А»
	После k-ой записи файла (со сдвигом)

	13
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Первую запись о выбранном виде транспорта без букв в номере маршрута
	С предыдущей записью файла

	14
	п
с
с
	Существо (вампир, оборотень и т.д.)

Имя

Место рождения
	Последнюю запись о существе с не менее чем двумя буквами «А» в имени
	С первой записью файла

	15
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Последнюю запись о выбранном виде транспорта с номером маршрута только из 2 и 3
	После последней записи файла (со сдвигом)

	16
	п
с
*
	Существо (вампир, оборотень и т.д.)

Имя

Дата рождения
	Первую запись о выбранном существе с именем на букву М
	С k-ой записью файла

	17
	с
ч
п
ч
	Событие

Число

Месяц (январь, февраль и т.д.)

Год
	Последнюю запись выбранного месяца с самым давним событием по дате
	С первой записью файла

	18
	п
с
*
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Дата утверждения маршрута
	Последнюю запись с самым недавним (по времени) утвержденным маршрутом
	Перед первой записью файла (со сдвигом)

	№
	Тип
	С1
	С2
	С3

	19
	с
п
п
ч
	Группа (например, А-07-14, ЭР-1-14 и т.д.)

Предмет (физика, матан и т.д.)

Оценка (отл, хор, удовл, неуд)

Количество студентов (из указанной Группы с указанной Оценкой по Предмету)
	Первую запись о группе с максимальным количеством выбранного значения оценки
	С предыдущей записью файла

	20
	с
п
с
	Преподаватель

Занятие (лекция, лаб, сем, экзамен и т.д.)

Аудитория (например, Ж-113, А-312 и т.д)
	Последнюю запись с введенной фамилией преподавателя и выбранным типом занятия
	После последней записи файла (со сдвигом)

	21
	п
с
*
	Существо (вампир, оборотень и т.д.)

Имя

Дата рождения
	Первую запись о выбранном существе с буквосочетанием «кус» в имени
	После k-ой записи файла (со сдвигом)

	22
	с,с,с
п
с
	ФИО

Тип (мобильный, городской, факс и т.д)

Телефонный номер
	Последнюю запись с выбранным типом связи с и именем Яков в любом регистре
	С предыдущей записью файла

	23
	с
ч
п
ч
	Событие

Число

Месяц (январь, февраль и т.д.)

Год
	Первую запись с событием на выбранную букву и самым недавним событием по дате
	С последней записью файла

	24
	п
с
*
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Дата утверждения маршрута
	Последнюю запись с самым давним (по времени) утвержденным маршрутом
	Перед k-ой записью файла (со сдвигом)

	25
	с,с,с
п
с
	ФИО

Тип (мобильный, городской, факс и т.д)

Телефонный номер
	Первую запись с введенным именем в любом регистре и номером, оканчивающимся на 0


	Со следующей записью файла

	26
	с
п
п
ч
	Группа (например, А-07-14, ЭР-1-14 и т.д.)

Предмет (физика, матан и т.д.)

Оценка (отл, хор, удовл, неуд)

Количество студентов (из указанной Группы с указанной Оценкой по Предмету)
	Первую запись о группе с буквой «Э» в названии с хотя бы одной хорошей оценкой по физике


	С k-ой записью файла

	27
	п
с
ч
*
	Тип (мобильный, городской, факс и т.д)

Телефонный номер

Сумма платежа

Дата платежа
	Первую запись с самым недавним (по дате) платежом за мобильную связь
	С предыдущей записью файла

	28
	п
с
ч
	Вид транспорта (автобус, трамвай и т.д.)
Номер маршрута (например, 10, 10А, А)

Количество остановок до ближайшего метро
	Первую запись о выбранном виде транспорта с буквой(амии) в номере маршрута


	После k-ой записи файла (со сдвигом)

	29
	п
с
ч
*
	Тип (мобильный, городской, факс и т.д)

Телефонный номер

Сумма платежа

Дата платежа
	Последнюю запись с самым давним (по дате) платежом на сумму менее 200
	Перед первой записью файла (со сдвигом)

	№
	Тип
	С1
	С2
	С3

	30
	п
с
*
	Существо (вампир, оборотень и т.д.)

Имя

Дата рождения
	Последнюю запись о существе с наибольшим количеством букв в имени
	Со следующей записью файла

	31
	п
с
ч
с
	Тип (мобильный, городской, факс и т.д)

Телефонный номер

Сумма платежа

Фамилия

	Первую запись с введенной фамилией в любом регистре суммой платеже не менее 300
	Перед k-ой записью файла (со сдвигом)

	32
	п
с
с
	Существо (вампир, оборотень и т.д.)

Имя

Место рождения
	Первую запись о выбранном существе с самым длинным именем
	С предыдущей записью 

	33
	с
п
с
	Преподаватель

Занятие (лекция, лаб, сем, экзамен и т.д.)

Аудитория (например, Ж-113, А-312 и т.д)
	Первую запись выбранным видом занятия и аудиторией, оканчивающейся на «13»
	Со следующей записью файла

	34
	с
п
п
ч
	Группа (например, А-07-14, ЭР-1-14 и т.д.)

Предмет (физика, матан и т.д.)

Оценка (отл, хор, удовл, неуд)

Количество студентов (из указанной Группы с указанной Оценкой по Предмету)
	Последнюю запись о группе с наибольшим количеством отличных оценок по выбранному предмету
	Со следующей записью файла

	35
	с,с,с
п
с
	ФИО

Тип (мобильный, городской, факс и т.д)

Телефонный номер
	Последнюю запись с номером телефона с не менее чем пятью цифрами «3»
	После последней записи файла (со сдвигом)


Пример программы с типизированным файлом

Задача. Найти первого Ивана c первого курса бакалавриата с хотя бы одной пятеркой из трёх оценок и изменить все его оценки на 3.
program PrjLAB2;  // два параметра – 1) имя типизир. файла, 2) имя текстового файла
{$APPTYPE CONSOLE}

uses

  SysUtils,     // ANSIUpperCase и др. -  смена регистра
  Windows,   // SetConsoleCP - смена кодировки
  UnLab2;   // свой модуль с типом и процедурами для решения задачи
//--------------------главная программа - меню -------------------------

var

  ch: char;

begin

  SetConsoleCP(1251);      // из модуля Windows

  SetConsoleOutputCP(1251);

  repeat

    writeln('--------------------------------------');
    writeln('D - создать новый тип.файл в диалоге; ');

    writeln('N - создать из текстового файла; ');

    writeln('F - поиск и изменение; ');

    writeln('V - просмотр;');

    writeln('E - конец.');

    write('Ваш выбор?'); readln(ch);

    writeln('--------------------------------------');
    ch:= UpCase(ch);  // символ в верхний регистр (латиница)
    case ch of

//--------------------------------------------------------------

//----------первая часть: создание типизированного в диалоге ---

    'D': CreateTypedFile1;  // вызов соответствующей процедуры для выбора D
//----------вторая часть: создание типизированного файла из текстового

    'N': CreateTypedFile2;

//----------третья часть: поиск и редактирование ---------------

    'F': FindIVAN;

//-----------просмотр----------------------------------------------

    'V': ViewFile;

//-----------выход----------------------------------------------

    'E': exit;

//--------------------------------------------------------------

      else

        begin

          writeln('Нет такой команды');

          write('Press ENTER'); readln;

        end;

    end;

  until ch='E';

end.

Исходный код модуля:
Unit  UnLab2;   // файл  UnLab2.pas

Interface

Uses
  SysUtils;     // ANSIUpperCase и др. -  смена регистра и обработка исключений
Type
  TKurs = (Bacalavr1, Bac2, Bac3, Bac4, Magistr1, Mag2); // с 0 до 5 по умолчанию
  TStud = record

    FIO: record

      F, I, O: string[30]; // Только статические строки можно сохранить в файл

      end;

    Kurs: TKurs;

    o1,o2,o3: byte;

  end;
//----------первая часть: создание типизированного файла в диалоге

Procedure CreateTypedFile1();

//----------вторая часть: создание типизированного файла из текстового

Procedure CreateTypedFile2();

//-------------- третья часть: поиск в типизированном файле ----

Procedure FindIVAN();

//-------------- просмотр типизированного файла ----

Procedure ViewFile();
Implementation // -----------реализационная часть-----------------
//----------первая часть: создание типизированного файла в диалоге

Procedure CreateTypedFile1();

var

  Stud: TStud;

  fr: file of TStud; // типизированный файл - файл записей

  n, kol: integer;

  ch: char;

  FileName:string[80];

begin

  if ParamCount<1 then

  begin

    writeln('Мало параметров '#13#10'Press enter');

    readln;  exit

  end;

  FileName := ParamStr(1);

// или простой диалог или через параметры программы при её вызове ParamStr(1)

// writeln('Введите имя типизированного файла');

// readln(FileName);

  AssignFile(fr, FileName);

  Try  ReWrite(fr);    //  или  {$I-} ReWrite(fr); {$I+} и if IOResult<>0 then <ошибка>
  Except

    writeln('Error: не удалось создать типизированный файл ',FileName);

    write('Press ENTER'); readln;  exit

  end;

  kol:=0;

  repeat

    write('Фамилия =?');   readln(Stud.FIO.F);

    write('Имя =?');       readln(Stud.FIO.I);

    write('Отчество =?');  readln(Stud.FIO.O);
// ввод значения перечисляемого типа (TKurs) через доп.переменную n целого типа
    write('Курс(1-6) =?'); readln(n); Stud.Kurs:=TKurs(n-1);  // 1-6 ( 0-5
    write('Три оценки: ');

    readln(Stud.o1, Stud.o2, Stud.o3);

    write(fr, Stud);   // запись записи целиком в файл

    inc(kol);

    write('Ещё? (y/n)');

    readln(ch)

  until UpCase(ch)='N';

  CloseFile(fr);

  writeln('Создан типизированный файл из ',kol,' записей');

  write('Press ENTER');

  readln;

end;
//---------------------------------------------------------------------------------------

//----------вторая часть: создание типизированного файла из текстового

Procedure CreateTypedFile2();

var

  Stud: TStud;

  ft: TextFile; // текстовый файл
  fr: file of TStud; // типизированный файл - файл записей
  n, kol: integer;

  ch: char;
  FileName:string[80];

begin

  if ParamCount<2 then

  begin

    writeln('Мало параметров '#13#10'Press enter');

    readln;  exit

  end;

FileName:= Paramstr(2);
 // или простой диалог или через параметры программы при её вызове ParamStr(2)

// writeln('Введите имя текстового файла');

// readln(FileName); 

 AssignFile(ft, FileName);

  Try  Reset(ft);

  Except

    writeln('Error: не удалось открыть текстовый файл ',Paramstr(2));

    write('Press ENTER'); readln;  exit

  end;

  //----------------- ЗАГЛУШКА --------------------

  //------------ Напишите код сами, а пока в диалоге: ----------------

              CreateTypedFile1; // заглушка - создание в диалоге –  заменить на свой код
  CloseFile(ft);

  write('Press ENTER');

  readln;

end;

//--------------------------------------------------------------

//-------------- третья часть: поиск в типизированном файле ----

Procedure FindIVAN();

var

  Stud: TStud;

  fr: file of TStud;

  n, kol: integer;

  FileName:string[80];

begin

  if ParamCount<1 then

  begin

    writeln('Мало параметров '#13#10'Press enter');

    readln;  exit

  end;

  FileName := ParamStr(1);

// или простой диалог или через параметры программы при её вызове ParamStr(1)

// writeln('Введите имя типизированного файла');

// readln(FileName); 
  AssignFile(fr, FileName);

  Try  ReSet(fr);    //  или  {$I-} ReSet(fr); {$I+} if IOResult<>0 then

  Except

    writeln('Error: не удалось открыть типизированный файл ',FileName);

    write('Press ENTER'); readln;  exit

  end;

  kol:=0;

  while not eof(fr) and (kol=0) do

  begin

    read(fr, Stud); // считывание
       // Если Иван c первого курса бакалавриата с хотя бы одной пятеркой
    if ((Stud.o1=5) or (Stud.o2=5) or (Stud.o3=5)) and

      (Stud.Kurs = Bacalavr1) and

      (ANSIUpperCase(Stud.FIO.I) = 'ИВАН') then  // сравнение в верхнем регистре
    begin

      writeln('Найден ', Stud.FIO.I,' ',Stud.FIO.O,' ',Stud.FIO.F,

      ' оценки ', Stud.o1, ' ', Stud.o2,' ', Stud.o3 );

      inc(kol);

      // Изменение считанного

      Stud.o1:=3; Stud.o2:=3; Stud.o3:=3;

      n:=FilePos(fr)-1; // позиция найденной записи – перед текущей
      // запись изменений "поверх" предыдущей версии

      seek(fr, n); // изменить тек.позицию в файле
      write(fr, Stud);       // запись в файле изменена

      writeln('В файле изменена запись номер ', n+1)

    end;

  end;

  if kol=0 then

    writeln('Данные, соответствующие запросу, не найдены');

  CloseFile(fr);

  write('Press ENTER');  readln;

end;

//--------------------------------------------------------------

//-------------- просмотр типизированного файла ----

Procedure ViewFile();

var

  Stud: TStud;

  fr: file of TStud;

  n, kol: integer;

  FileName:string[80];

begin

  if ParamCount<1 then

  begin

    writeln('Мало параметров '#13#10'Press enter');

    readln;  exit

  end;

  FileName := ParamStr(1);

// или простой диалог или через параметры программы при её вызове ParamStr(1)

// writeln('Введите имя типизированного файла');

// readln(FileName); 

  AssignFile(fr, FileName);

  Try  ReSet(fr);    //  или  {$I-} ReSet(fr); {$I+} if IOResult<>0 then

  Except

    writeln('Error: не удалось открыть типизированный файл ',FileName);

    write('Press ENTER'); readln;  exit

  end;

  kol:=0;

  while not eof(fr) do

  begin

    read(fr, Stud); // считывание
      writeln('Найден ', Stud.FIO.I,' ',Stud.FIO.O,' ',Stud.FIO.F,

      ' курс ', Ord(Stud.Kurs)+1,

      ' оценки ', Stud.o1, ' ', Stud.o2,' ', Stud.o3 );

      inc(kol);

  end;

  if kol=0 then

    writeln('Данные не найдены')

  else

    writeln('Найдено ',kol, ' записей');

  CloseFile(fr);

  write('Press ENTER');  readln;

end;

End.
Контрольные вопросы
1. Как можно открыть типизированный файл? Текстовый файл?

2. В чем отличия процедуры ReSet для открытия текстового и типизированного файлов?

3. Есть ли отличия считывания записи из текстового и типизированного файлов?

4. Есть ли отличия записывания записи в текстовый и типизированный файлы?

5. Можно ли дописать информацию в конец текстового файла? Типизированного файла? Как?

6. Можно ли дописать информацию в начало текстового файла? Типизированного файла? Как?

7. Допустим, в типизированном файле находятся три записи. Если перейти в начало файла и записать туда одну запись, сколько записей будет в файле – три или четыре? Почему?
